Процедури от по-висок ред

Сума

$$\sum_{x=a}^{b} x$$

Сума... на целите числа

$$\sum_{x=a}^{b} x$$

```
(define (sum-integers a b)
  (if (> a b)
      0
      (+ a
         (sum-integers (+ a 1) b))))
```

Сума... на квадратите

$$\sum_{x=a}^{b} x^2$$

```
(define (sum-squares a b)
  (if (> a b)
      0
      (+ ...
         (sum-squares (+ a 1) b))))
```

Сума... на квадратите

$$\sum_{x=a}^{b} x^2$$

```
(define (sum-squares a b)
  (if (> a b)
      0
      (+ (square a)
         (sum-squares (+ a 1) b))))
```

Сума... на дроби

$$\sum_{x=a}^{b} \frac{1}{x^4 + 1}$$

```
(define (sum-fractions a b)
  (if (> a b)
      0
      (+ ...
         (sum-fractions (+ a 1) b))))
```

Сума... на дроби

$$\sum_{x=a}^{b} \frac{1}{x^4 + 1}$$

```
(define (sum-fractions a b)
  (if (> a b)
      0
      (+ (/ 1
            (+ (square (square a)) 1))
         (sum-fractions (+ a 1) b))))
```

Обобщение

$$\sum_{x=a}^{b} x \qquad \sum_{x=a}^{b} x^2 \qquad \sum_{x=a}^{b} \frac{1}{x^4+1} \qquad \cdots$$

Обобщение

$$\sum_{x=a}^{b} x \qquad \sum_{x=a}^{b} x^2 \qquad \sum_{x=a}^{b} \frac{1}{x^4+1} \qquad \cdots \qquad \sum_{x=a}^{b} f(x)$$

По-абстрактна сума

$$\sum_{x=a}^{b} f(x)$$

```
(define (sum f a b)
  (if (> a b)
      0
      (+ (f a)
         (sum f (+ a 1) b))))
```

По-абстрактна сума

$$\sum_{x=a}^{b} term(x)$$

```
(define (sum term a b)
  (if (> a b)
      0
      (+ (term a)
         (sum term (+ a 1) b))))
```

По-абстрактна сума, с приложения

```
(define (identity x) x)

(define (sum-integers a b)
  (sum identity a b))
```

```
(define (square x) (* x x))

(define (sum-squares a b)
  (sum square a b))
```

```
(define (cube x) (* x x x))

(define (sum-cubes a b)
  (sum cube a b))
```

```
(define (sum-fractions a b)
  (define (term x)
    (/ 1
       (+ (square (square x)) 1)))

  (sum term a b))
```

Произведение

$$\prod_{x=a}^{b} f(x)$$

Произведение

$$\prod_{x=a}^{b} term(x)$$

```
(define (product term a b)
  (if (> a b)
      0
      (* (term a)
         (product term (+ a 1) b))))
```

$$\sum_{x=a}^{b} f(x) \qquad \prod_{x=a}^{b} f(x) \qquad \bigwedge_{x=a}^{b} f(x) \qquad \cdots$$

Още по-абстрактно - ние просто акумулираме

$$\sum_{x=a}^{b} f(x) \qquad \prod_{x=a}^{b} f(x) \qquad \bigwedge_{x=a}^{b} f(x) \qquad \cdots$$

```
(define (accumulate combine term a b)
  (if (> a b)
      0
      (combine (term a)
               (accumulate combine
                           term
                           (+ a 1)
                           b))))
```

Но виждате ли проблем?

(* Произведение 0) ?

Нека дефинираме `product` в термините на `accumulate`:

```
(define (product term a b)
  (accumulate * term a b))
```

И нека приложим `product`:

```
(define (identity x) x)

(product identity 1 5) ; 0
```

```
(define (square x) (* x x))

(product square 2 3) ; 0
```

```
(define (identity x) x)

(define (factorial n)
  (product identity 1 n))
```

(* Произведение 1)

Трябва да зададем началната стойност на нашия акумулатор

Още по-абстрактно

$$\sum_{x=a}^{b} f(x) \qquad \prod_{x=a}^{b} f(x) \qquad \bigwedge_{x=a}^{b} f(x) \qquad \cdots$$

```
(define (accumulate combine null-value term a b)
  (if (> a b)
      null-value
      (combine (term a)
               (accumulate combine
                           null-value
                           term
                           (+ a 1)
                           b))))
```

От частното към общото



$$a^2 + b^2 = c^2$$

$$c^2 = a^2 + b^2 - 2ab\cos\gamma$$

Let H be a (real) Hilbert space,
$\|.\|$ be the by the inner product induced norm.
Then we have $\forall x, y \in H$:

$$\|x + y\|^2 = \|x\|^2 + \|y\|^2 + 2\langle x, y\rangle$$

if A is an $n \times k$ matrix, then

$$det(A^t A) = \sum_I det(A_I)^2$$

where the sum is over all $k \times k$ minors of A.

Да дефинираме `sum` и `product` чрез `accumulate`

```
(define (sum term a b)
  (accumulate + 0 term a b))

(define (product term a b)
  (accumulate * 1 term a b))
```

Или нещо по-конкретно

```
(define (square x) (* x x))

(define (sum-squares a b)
  (accumulate + 0 square a b))
```

```
(define (cube x) (* x x x))

(define (product-cubes a b)
  (accumulate * 1 cube a b))
```

```
(define (average x y) ; бинарна процедура
  (/ (+ x y) 2))

(define (identity x) x)

(define (fractions a b)
  (accumulate average 0 identity a b))
```

λ

Създаване на процедури чрез `lambda`

```
(define (square x) (* x x))
```

```
(define square (lambda (x) (* x x)))
```

```
(sum 1 5 square) ; 55
```

```
(sum 1 5 (lambda (x) x)) ; 15
```

λ

Как да четем lambda

```
(lambda                        (x)      (+   x      4))
    |                           |        |   |      |
the procedure of an argument x that adds x and 4
```

```
(define (square    x)          (*      x          x))
  |       |       |            |      |          |
To     square  something,   multiply  it  by  itself.
```